```
> with(OreModules):
> with(OreMorphisms):
> with(Stafford):
> with(linalg):
```

Let us consider the second Weyl algebra $A = A_2(\mathbb{Q})$, where $\mathbb{Q}$ is the field of rational numbers,

```
> A := DefineOreAlgebra(diff=[d[1],x[1]], diff=[d[2],x[2]],
> polynom=[x[1],x[2]]):
```

and the left A-module M finitely presented by the matrix R defined by:

```
> R := evalm([[0,d[1],d[2]+x[1]]]);
```

$$R := \begin{bmatrix} 0 & d_1 & d_2 + x_1 \end{bmatrix}$$

The rank of the left A-module M is:

```
> OreRank(R,A);
```

$$2$$

Since R admits a right inverse S defined by

```
> S := RightInverse(R,A);
```

$$S := \begin{bmatrix} 0 \\ d_2 + x_1 \\ -d_1 \end{bmatrix}$$

M is a stably free left A-module of rank 2, i.e., a free left A-module of rank 2. Using the fact that the direct sum of M and A is isomorphic to $A^{1\times 3}$, and using the Cancellation Theorem, let us compute a basis of M. Let us first compute

```
> X := stackmatrix(R,1-Mult(S,R,A));
```

$$X := \begin{bmatrix} 0 & d_1 & d_2 + x_1 \\ 1 & 0 & 0 \\ 0 & -(d_2 + x_1)\, d_1 + 1 & -(d_2 + x_1)^2 \\ 0 & {d_1}^2 & 2 + d_1\, d_2 + d_1\, x_1 \end{bmatrix}$$

which defines the left A-isomorphism g from the direct sum of A and M onto $A^{1\times 3}$. Moreover, the direct sum of A and M is isomorphic to the left A-module L finitely presented by the matrix P defined by

```
> P := augment(evalm([[0]]),R);
```

$$P := \begin{bmatrix} 0 & 0 & d_1 & d_2 + x_1 \end{bmatrix}$$

Similarly, a finite presentation of $A^{1\times 2}$ is given by the matrix R' defined by

```
> Rp := evalm([[0$2]]);
```

$$Rp := \begin{bmatrix} 0 & 0 \end{bmatrix}$$

and $A^{1\times 3}$ is isomorphic to the left A-module L' finitely presented by the matrix P' = (0 R') defined by

```
> Pp := augment(evalm([[0]]),Rp);
```

$$Pp := \begin{bmatrix} 0 & 0 & 0 \end{bmatrix}$$

Let us check again that the left A-homomorphism f from L to L' induced by X is a left A-isomorphism:

```
> TestIso(P,Pp,X,A);
```

$$true$$

We can apply the algorithm corresponding to the Cancellation Theorem to P, P', and X

```
> Q := Cancellation(Rp,X,A,"splithom");
```

$$Q := \begin{bmatrix} d_1 & d_2 + x_1 \\ -(d_2 + x_1)\, d_1 + 1 & -(d_2 + x_1)^2 \\ {d_1}^2 & 2 + d_1\, d_2 + d_1\, x_1 \end{bmatrix}$$

to obtain a left A-isomorphism h from M onto $A^{1\times 2}$ induced by Q. Let us check again that h is a left A-isomorphism:

```
> TestIso(R,Rp,Q,A);
```

$$true$$

Thus, the matrix Q defines an injective parametrization of M, i.e., we have $ker_A(.\mathrm{Q}) = A\ \mathrm{R}$

```
> SyzygyModule(Q,A);
```

$$\begin{bmatrix} 0 & d_1 & d_2 + x_1 \end{bmatrix}$$

and Q admits a left inverse

```
> T := LeftInverse(Q,A);
```

$$T := \begin{bmatrix} d_2 + x_1 & 1 & 0 \\ -d_1 & 0 & 1 \end{bmatrix}$$

Hence, we get the exact sequence

$$0 \longrightarrow A \xrightarrow{.R} A^{1\times 3} \xrightarrow{.Q} A^{1\times 2} \longrightarrow 0,$$

which shows that M, which is the cokernel of .R, is isomorphic to $im_A(.\mathrm{Q}) = A^{1\times 3}\, Q = A^{1\times 2}$. In particular, the residue classes of the rows of the left inverse T of Q define a basis of M.

Let us compare this approach with the ones developed in F. Chyzak, A. Quadrat, D. Robertz, "Effective algorithms for parametrizing linear control systems over Ore algebras", Appl. Algebra Engrg. Comm. Comput. 16 (2005), pp. 319-376 and A. Quadrat, D. Robertz, "Computation of bases of free modules over the Weyl algebras", J. Symbolic Comput. 42 (2007), pp. 1113-1141. Let us first check if we can compute a basis of M by means of a minimal parametrization.

```
> K := MinimalParametrizations(R,A);
```

$$K := \left[\begin{bmatrix} 1 & 0 \\ 0 & -{d_2}^2 - 2\, d_2\, x_1 - {x_1}^2 \\ 0 & 2 + d_1\, d_2 + d_1\, x_1 \end{bmatrix}, \begin{bmatrix} 1 & 0 \\ 0 & -d_1\, d_2 - d_1\, x_1 + 1 \\ 0 & {d_1}^2 \end{bmatrix} \right]$$

We obtain that $ker_A(.\mathrm{K[i]}) = A\ \mathrm{R}$ for i = 1, 2, and thus, M is isomorphic to $A^{1\times 3}$ K[i]. But none of the K[i]'s admits a left inverse.

```
> map(LeftInverse,K,A);
```

$$[[], []]$$

Therefore, we get that $A^{1\times3}$ K[i] is a proper left A-submodule of $A^{1\times2}$, and thus these two minimal parametrizations do not define an injective parametrization of M.

Let us now use the algorithm developed in A. Quadrat, D. Robertz, "Computation of bases of free modules over the Weyl algebras", J. Symbolic Comput. 42 (2007), pp. 1113-1141, and implemented in the Stafford package for the computation of bases of free left A-modules of rank greater than or equal to 2:

```
> Qp := InjectiveParametrization(R,A);
```

$$Qp := \begin{bmatrix} d_2 + x_1 & d_1 \\ d_1\, {d_2}^2 + 2\, d_1\, d_2\, x_1 - {d_2}^2 - 2\, d_2\, x_1 + d_1\, {x_1}^2 - {x_1}^2 & -d_1 + {d_1}^2\, d_2 + {d_1}^2\, x_1 + 1 - d_1\, d_2 - d_1\, x_1 \\ 2 - {d_1}^2\, d_2 + d_1\, d_2 - {d_1}^2\, x_1 - 2\, d_1 + d_1\, x_1 & -{d_1}^3 + {d_1}^2 \end{bmatrix}$$

We obtain that ker_A(.Q') $=$ A R

```
> SyzygyModule(Qp,A);
```

$$\begin{bmatrix} 0 & d_1 & d_2 + x_1 \end{bmatrix}$$

and Q' admits a left inverse T' defined by

```
> Tp := LeftInverse(Qp,A);
```

$$Tp := \begin{bmatrix} {d_1}^2 - d_1 & 0 & 1 \\ -d_1\, d_2 - d_1\, x_1 + d_2 + x_1 + 1 & 1 & 0 \end{bmatrix}$$

which yields that M is isomorphic to $A^{1\times2}$ Q' $=$ $A^{1\times2}$.